



## 第四章 语法分析

1. 语法分析的功能、基本任务
2. 自顶向下分析法
3. 自底向上分析法



## 4.1 语法分析概述

**功能:** 根据文法规则, 从源程序单词符号串中识别出语法成分, 并进行语法检查。

**基本任务:** 识别符号串 $S$ 是否为某语法成分。

两大类分析方法:

自顶向下分析

自底向上分析



自顶向下分析算法的基本思想为：

$$\text{若 } Z \xrightarrow[G[Z]}{+} S \text{ 则 } S \in L(G[Z]) \text{ 否则 } S \notin L(G[Z])$$

? 存在主要问题：

- 左递归问题
- 回溯问题

■ 主要方法：

- 递归子程序法
- LL分析法



自底向上分析算法的基本思想为：

若  $Z \leftarrow S$  则  $S \in L(G[Z])$  否则  $S \notin L(G[Z])$

? 存在主要问题：

- 句柄的识别问题
- 若两个以上规则的右部有相同的符号串且构成句柄，选哪个候选式？

主要方法：

- 算符优先分析法
- LR分析法



# 语法分析部分（知识点总图）

	存在主要问题	主要方法
自顶向下	<ul style="list-style-type: none"><li>1. 左递归</li><li>2. 回溯</li></ul>	<ul style="list-style-type: none"><li>1. 递归子程序法</li><li>2. LL分析法</li></ul>
自底向上	<ul style="list-style-type: none"><li>1. 如何识别句柄</li><li>2. 若两个以上规则的右部都能构成句柄时，选哪个</li></ul>	<ul style="list-style-type: none"><li>1. 算符优先分析法</li><li>2. LR分析法</li></ul>



## 4.2 自顶向下分析

### 4.2.1 自顶向下分析的一般过程

给定符号串  $S$ ，若预测它是某一语法成分，那么可根据该语法成分的文法，设法为  $S$  构造一棵语法树。

若成功，则  $S$  最终被识别为某一语法成分。即

$S \in L(G[Z])$  其中  $G[Z]$  为某语言成分的文法。

若不成功，则  $S \notin L(G[Z])$ 。

- 我们可以通过一例子来说明语法分析过程



例

$S = c a d$

$G[Z]:$

$Z ::= c A d$

$A ::= a b | a$

求解  $S \in L(G[Z]) ?$

2型文法!  
上下文无关文法

分析过程是设法建立一棵语法树，使语法树的末端结点与给定符号串相匹配。


1. 开始：令Z为根结点。

2. 用Z的右部符号串去匹配输入串

完成一步推导  $Z \Rightarrow c A d$

检查  $c - c$  匹配

A是非终结符，将匹配任务交给A





$$S = c \text{ a } d \quad G[Z]: Z ::= c A d \\ A ::= a b \mid a$$

3. 选用A的右部符号串匹配输入串  
A有两个右部，选第一个。

完成进一步推导  $A \Rightarrow a b$



检查：a - a匹配，b - d不匹配（失败）

但是还不能贸然宣布  $S \notin L(G[Z])$

4. 回溯：即砍掉A的子树  
改选A的第二个右部

$A \Rightarrow a$  检查 a - a匹配  
d - d匹配

语法树末端结点为c a d与输入c a d相匹配，建立了推导序列  $Z \Rightarrow c A d \Rightarrow c a d$ 。  
 $\therefore c a d \in L(G(Z))$





## 自顶向下分析方法的特点

1. 分析过程是带有预测的。即要根据输入符号串中下一个单词，来预测之后的内容属于什么语法成分，然后用相应语法成分的文法建立语法树。
2. 分析过程是一种试探过程，是尽一切办法（选用不同规则）设法建立语法树的过程。由于是试探过程，故难免有失败，所以分析过程需进行回溯，因此我们也称这种方法是带回溯的自顶向下分析方法。
3. 最左推导可以编出程序来实现，但在实际上价值不大，效率低，代价高。



## 4.2.2 自顶向下分析存在的问题及解决方法

### 一、左递归文法：

有如下文法：

令U是文法的任一非终结符，文法中有规则

$U ::= U \dots$  或者  $U \Rightarrow U \dots$

这个规则文法是左递归的。

该方法的基本缺点是不能处理具有左递归性的文法。

自顶向下分析为什么不能处理左递归文法？



如果我们在匹配输入串过程中，假定正好轮到要用非终结符U直接匹配输入串，即要用U的右部符号串U……去匹配，为了用U……去匹配，又得用U去匹配，这样无限地循环下去，过程将无法终止。

如果文法具有间接左递归，则也将发生上述问题，只不过兜的圈子更大。

要实行自顶向下分析，必须要消除文法的左递归，下面我们将介绍直接左递归的消除方法。在此基础上再介绍一般左递归的消除方法。



## 消除直接左递归

方法一：使用扩充的BNF表示来改写文法

例： (1)  $E ::= E + T \mid T \Rightarrow E ::= T \{ + T \}$   
(2)  $T ::= T * F \mid T / F \mid F \Rightarrow T ::= F \{ * F \mid / F \}$

- a. 改写以后的文法消除了左递归。
- b. 可以证明，改写前后的文法是等价的，表现在

$$L(G_{\text{改前}}) = L(G_{\text{改后}})$$

如何改写文法能消除左递归，又前后等价？  
现在我们可以给出两条规则。



## 规则一（提因子）

注意若是  $y \ x \mid w \ x \mid \dots \mid z \ x$   
则  $x$  不是公因子！

若：  $U ::= x \ y \mid x^{\bullet} w \mid \dots \mid x \ z$

则可改写为：  $U ::= x ( y \mid w \mid \dots \mid z )$

其中再若：  $y = y_1 y_2, \quad w = w_1 w_2$

则  $U ::= x ( y_1 ( y_2 \mid w_2 ) \mid \dots \mid z )$

若有规则：  $U ::= x \mid x \ y$

则可以改写为：  $U ::= x ( y \mid \epsilon )$

注意： 不应写成  $U ::= x (\epsilon \mid y)$

总把  $\epsilon$  安置成最后的选择！

使用提因子法，不仅有助于消除直接左递归，而且有助于压缩文法的长度，使我们更加有效地分析句子。



## 规则二

若有文法规则:  $U ::= x \mid y \mid \dots \mid z \mid U v$

其特点是: 具有一个直接左递归的右部并位于最后,  
这表明该语法类  $U$  是由  $x$  或  $y \dots$  或  $z$  打头, 其后跟随  
零个或多个  $v$  组成。

$$U \Rightarrow Uv \Rightarrow Uvv \Rightarrow Uvvv \Rightarrow \dots$$

$\therefore$  可以改写为  $U ::= (x \mid y \mid \dots \mid z) \{ v \}$

通过以上两条规则, 就能消除文法的直接左递归, 并  
且保证文法的等价性。



## 方法二：将左递归规则改为右递归规则

### 规则三

若:  $P ::= P\alpha \mid \beta$

则可改写为:  $P ::= \beta P'$

$P' ::= \alpha P' \mid \epsilon$

下面有两个例题:



例1  $E ::= E + T | T$

产生式右部无公因子, 所以不能用规则一。

为了使用规则二, 我们

令  $E ::= T | E + T$

$\therefore$  由规则二我们可以得到

$$E ::= T \{ + T \}$$

右递归:

$$E ::= T E'$$

$$E' ::= + T E' | \epsilon \quad \text{规则三}$$

例2  $T ::= T * F | T / F | F$

$$T ::= F | T * F | T / F$$

$$T ::= F | T (* F | / F) \quad \text{规则一}$$

$$T ::= F \{ (* F | / F) \} \quad \text{规则二}$$

$$\text{即 } T ::= F \{ * F | / F \}$$

右递归:

$$T ::= F T'$$

$$T' ::= * F T' | / F T' | \epsilon \quad \text{规则三}$$



## 消除一般左递归

一般左递归也可以通过改写法予以消除。

消除所有左递归的算法：

1. 把G的非终结符整理成某种顺序  $A_1, A_2, \dots, A_n$ ，使得

$A_1 ::= \delta_1 | \delta_2 | \dots | \delta_k$

$A_2 ::= A_1 r \dots$

$A_3 ::= A_2 u | A_1 v \dots$

.....



2. **for**  $i := 1$  **to**  $n$  **do**

**begin**

**for**  $j := 1$  **to**  $i-1$  **do**

把每个形如  $A_i ::= A_j r$  的规则替换成

$A_i ::= (\delta_1 | \delta_2 | \dots | \delta_k) r$

其中  $A_j ::= \delta_1 | \delta_2 | \dots | \delta_k$  是当前  $A_j$  的全部规则

消除  $A_i$  规则中的直接左递归

**end**

3. 化简由 2 得到的文法，即去掉那些多余的规则。



例：文法  $G[S]$  为

$$S ::= Q c \mid c$$

$$Q ::= R b \mid b$$

$$R ::= S a \mid a$$

该文法无直接左递归，但有间接左递归

$$S \Rightarrow Q c \Rightarrow R b c \Rightarrow S a b c \quad \therefore S \not\Rightarrow S a b c$$

非终结符顺序重新排列

$$R ::= S a \mid a$$

$$Q ::= R b \mid b$$

$$S ::= Q c \mid c$$



$R ::= S \ a \mid a$   
 $Q ::= R \ b \mid b$   
 $S ::= Q \ c \mid c$

1. 检查规则R是否存在直接左递归       $R ::= S \ a \mid a$

2. 把R代入Q的有关选择， 改写规则Q       $Q ::= S \ a \ b \mid a \ b \mid b$

3. 检查Q是否直接左递归

4. 把Q代入S的右部选择       $S ::= S \ a \ b \ c \mid a \ b \ c \mid b \ c \mid c$

5. 消除S的直接左递归       $S ::= ( a \ b \ c \mid b \ c \mid c ) \{ a \ b \ c \}$



最后得到文法为：

$S ::= (a b c | b c | c) \{ a b c \}$

$Q ::= S a b | a b | b$

$R ::= S a | a$

可以看出其中关于Q和R的规则是多余的规则

$\therefore$  经过压缩后  $S ::= (a b c | b c | c) \{ a b c \}$

可以证明改写前后的文法是等价的

应该指出，由于对非终结符的排序不同，最后得到的文法在形式上可能是不一样的，但是不难证明它们的等价性。



## 二、回溯问题

什么是回溯？

分析工作要部分地或全部地退回去重做叫回溯。

造成回溯的条件：

$$U ::= \alpha_1 \mid \alpha_2 \mid \alpha_3$$

文法中，对于某个非终结符号的规则其右部有多个选择，并根据所面临的输入符号不能准确地确定所要的产生式，就可能出现回溯。

回溯带来的问题：

效率严重低下，只有在理论上的意义而无实际意义。



## 效率低的原因

- 1) 语法分析要重做
- 2) 语法处理工作要推倒重来

## 怎么样才能避免回溯？

[定义] 设文法  $G$  (不具左递归性),  $U \in V_n$

$$U ::= \alpha_1 \mid \alpha_2 \mid \alpha_3$$

$$\text{FIRST}(\alpha_i) = \{a \mid \alpha_i \Rightarrow a \dots, a \in V_t\}$$

为避免回溯, 对文法的要求是:

$$\text{FIRST}(\alpha_i) \cap \text{FIRST}(\alpha_j) = \emptyset \quad (i \neq j)$$



## 消除回溯的途径:

### 1. 改写文法

对具有多个右部的规则**反复提取左因子**

例1  $U ::= x V \mid x W$

$U, V, W \in V_n, x \in V_t$

改写为  $U ::= x (V \mid W)$

更清楚表示

$U ::= x Z$

$Z ::= V \mid W$

注意：问题到此并没有结束，还需要进一步检查V和W的首符号是否相交

若  $V ::= a b \mid c d$        $\text{FIRST}(V) = \{ a, c \}$

$W ::= d e \mid f g$        $\text{FIRST}(W) = \{ d, f \}$

只要不相交就可以根据输入符号确定目标；若相交，则要代入，并再次提取左因子。如：  $V ::= a b$      $W ::= a c$

则：  $Z ::= a (b \mid c)$



## 例2: 文法G[<程序>]

<程序> ::= <分程序> | <复合语句>

<分程序> ::= begin<说明串>; <语句串> end

<复合语句> ::= begin<语句串> end

$\text{FIRST}(<\text{分程序}>) = \{ \text{begin} \}$

$\text{FIRST}(<\text{复合语句}>) = \{ \text{begin} \}$

改写文法:

<程序> ::= begin (<说明串>; <语句串> end | <语句串> end)

引入 <程序'>

<程序> ::= begin <程序'>

<程序'> ::= <说明串>; <语句串> end | <语句串> end



```
<程序> ::= begin <程序’>
<程序’> ::= <说明串>; <语句串> end | <语句串> end
```

对于: <程序’>

**FIRST( <说明串>; <语句串> end )**

= { real, integer, boolean, array, function, procedure }

**FIRST( <语句串> end )**

= { 标识符, goto, begin, if, for }

不相交。



## 2. 超前扫描

当文法不满足避免回溯的条件时，即各选择的首符号相交时，可以采用超前扫描的方法，即向前侦察各输入符号串的第二个、第三个符号来确定要选择的目标。

这种方法是通过向前多看几个符号来确定所选择的目标，从本质上来说也有回溯的味道，因此比第一种方法费时，但是读的仅仅是向前侦察情况，不作任何语义处理工作。



例

$\langle \text{程序} \rangle ::= \langle \text{分程序} \rangle \mid \langle \text{复合语句} \rangle$

$\langle \text{分程序} \rangle ::= \text{begin } \langle \text{说明串} \rangle; \langle \text{语句串} \rangle \text{ end}$

$\langle \text{复合语句} \rangle ::= \text{begin } \langle \text{语句串} \rangle \text{ end}$

这两个选择的首符号是相交，因此读到begin时并不能确定该用哪个选择。这时可采用向前假读进行侦察，此例题只需假读一次就可以确定目标。

因为 $\langle \text{说明串} \rangle$ 的首符集为{real, integer, ...., procedure}; 而 $\langle \text{语句串} \rangle$ 的首符集为{标识符, if, for, ...., begin}。

$\therefore$ 只要超前假读得到的是“说明串”的首符，便是第一个选择。若是“语句串”的首符，就是第二个选择。



## 文法的两个条件

为了在不采取超前扫描的前提下实现不带回溯的自顶向下分析，对文法需要满足两个条件：

1. 文法是非左递归的；
2. 对文法的任一非终结符，若其规则右部有多个选择时，各选择所推出的终结符号串的首符号集合要两两不相交。

在上述条件下，就可以根据文法构造有效的、不带回溯的自顶向下分析器。



### 4.2.3 递归子程序法（递归下降分析法）

具体做法：对语法的每一个非终结符都编一个分析程序。当根据文法和当时的输入符号预测到要用某个非终结符去匹配输入串时，就调用该非终结符的分析程序。

下面我们可以通过举例说明如何根据文法构造该文法的语法分析程序。




如文法  $G[E]$ :  $E ::= U V$


$U ::= \dots$

$V ::= \dots$


$E$  的分析程序



$U$  的分析程序



$V$  的分析程序



注: 消除左递归后, 可有其它递归 (如右递归或自嵌入递归):

$U ::= \dots U \dots$

$U ::= \dots W \dots$

$W ::= \dots U \dots$



## 例：文法G[Z]

$$\begin{aligned}Z &::= (U) \mid aUb \\U &::= dZ \mid Ud \mid e\end{aligned}$$

注意：这两个式子中的圆括号不同！  
Z中的括号是终结符号之一。  
U中的括号则是元语言的括号。

### 1. 检查并改写文法

$$\begin{aligned}Z &::= (U) \mid aUb \\U &::= (dZ \mid e) \{ d \}\end{aligned}$$

改写后无左递归且首符集不相交：

$$\begin{aligned}\{( \} \cap \{ a \} &= \emptyset \\ \{ d \} \cap \{ e \} &= \emptyset\end{aligned}$$

### 2. 检查文法的递归性

$$Z \Rightarrow \dots U \dots \Rightarrow \dots Z \dots$$

$$\therefore Z \not\Rightarrow^* \dots Z \dots$$

$$U \Rightarrow \dots Z \dots \Rightarrow \dots U \dots$$


$$\therefore U \not\Rightarrow^* \dots U \dots$$

因此，Z和U的分析程序可以编成递归子程序




### 3. 算法框图

$Z ::= ( U ) \mid a U b$





$U ::= (d \ Z \mid e) \{ d \}$





## 说明：

- 非终结符号的分析子程序功能，是用规则右部符号串去匹配输入串。
- 要注意子程序之间的接口。在程序编制时，进入某个非终结符的分析程序前，其所要分析的语法成分的第一个符号已读入sym中了。
- 递归子程序法对应的是最左推导过程！
- 在上面的分析过程中，我们强调一定要消除左递归，但允许存在右递归或自嵌入递归。正是由于在子程序的调用过程中允许（直接或间接的）递归调用，这种方法才得名递归子程序法。



## 4.2.4 用递归子程序法构造语法分析程序的例子

文法:     $\langle \text{语句} \rangle ::= \langle \text{变量} \rangle := \langle \text{表达式} \rangle$

          | IF  $\langle \text{表达式} \rangle$  THEN  $\langle \text{语句} \rangle$

          | IF  $\langle \text{表达式} \rangle$  THEN  $\langle \text{语句} \rangle$  ELSE  $\langle \text{语句} \rangle$

$\langle \text{变量} \rangle ::= i | i [ \langle \text{表达式} \rangle ]$

$\langle \text{表达式} \rangle ::= \langle \text{项} \rangle | \langle \text{表达式} \rangle + \langle \text{项} \rangle$

$\langle \text{项} \rangle ::= \langle \text{因子} \rangle | \langle \text{项} \rangle * \langle \text{因子} \rangle$

$\langle \text{因子} \rangle ::= \langle \text{变量} \rangle | ( \langle \text{表达式} \rangle )$

用单引号括起来的  
符号表示终结符号

改写文法:     $\langle \text{语句} \rangle ::= \langle \text{变量} \rangle := \langle \text{表达式} \rangle$

          | IF  $\langle \text{表达式} \rangle$  THEN  $\langle \text{语句} \rangle$  [ ELSE  $\langle \text{语句} \rangle$  ]

$\langle \text{变量} \rangle ::= i [ [ \langle \text{表达式} \rangle ] ]$

$\langle \text{表达式} \rangle ::= \langle \text{项} \rangle \{ + \langle \text{项} \rangle \}$

$\langle \text{项} \rangle ::= \langle \text{因子} \rangle \{ * \langle \text{因子} \rangle \}$

$\langle \text{因子} \rangle ::= \langle \text{变量} \rangle | ( \langle \text{表达式} \rangle )$



语法分析程序所要调用的子程序：

**nextsym:** 词法分析程序，每调用一次读进一个单词，  
单词的类别码放在sym中。

**error:** 出错处理程序。



```
PROCEDURE state; /*语句*
  IF sym = 'IF' THEN
    BEGIN nextsym; expr;
      IF sym ≠ 'THEN' THEN error
      ELSE BEGIN nextsym; state; END

      IF sym = 'ELSE' THEN
        BEGIN
          nextsym;
          state;
        END
      END
    ELSE BEGIN var;
      IF sym ≠ ':=' THEN error
      ELSE BEGIN
        nextsym;
        expr;
      END
    END
  END
```



<变量> ::= i [ '<表达式>' ]

```
PROCEDURE var;          /*变量*
IF sym ≠ 'i' THEN error
ELSE BEGIN
    nextsym;
    IF sym = '[' THEN
        BEGIN
            nextsym;
            expr;
            IF sym ≠ ']' THEN error
            ELSE nextsym;
        END
    END
```



<表达式> ::= <项> { + <项> }

```
PROCEDURE expr;          /*表达式*/  
BEGIN term;  
  WHILE sym = '+' DO  
    BEGIN  
      nextsym;  
      term;  
    END  
  END
```



<项> ::= <因子> { \* <因子> }  
<因子> ::= <变量> | '(' <表达式> ')'

```
PROCEDURE term; /*项*/  
BEGIN factor;  
WHILE sym = '*' DO  
BEGIN nextsym;  
factor;  
END  
END
```

---

```
PROCEDURE factor; /*因子*/  
BEGIN  
IF sym='(' THEN  
BEGIN nextsym; expr;  
IF sym != ')' THEN error  
ELSE nextsym  
END  
ELSE var;  
END
```



## 举例分析

if ( i + i ) then i := i \* i + i else

i [ i ] := i + i [ i \* i ] \* ( i + i )

作业：

p84: 1

p91: 1-3

【注意：第3题第3个产生式应为B ::= aB|a而不是B ::= aA|a】